NAG Library Routine Document

G01BKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01BKF returns the lower tail, upper tail and point probabilities associated with a Poisson distribution.

2 Specification

```
SUBROUTINE G01BKF (RLAMDA, K, PLEK, PGTK, PEQK, IFAIL)
INTEGER            K, IFAIL
REAL (KIND=nag_wp) RLAMDA, PLEK, PGTK, PEQK
```

3 Description

Let X denote a random variable having a Poisson distribution with parameter λ (> 0). Then

$$\text{Prob}\{X = k\} = e^{-\lambda}\frac{\lambda^k}{k!}, \quad k = 0, 1, 2, \ldots$$

The mean and variance of the distribution are both equal to λ.

G01BKF computes for given λ and k the probabilities:

$$\begin{aligned}\text{PLEK} &= \text{Prob}\{X \leq k\}\\ \text{PGTK} &= \text{Prob}\{X > k\}\\ \text{PEQK} &= \text{Prob}\{X = k\}.\end{aligned}$$

The method is described in KnÏsel (1986).

4 References

KnÏsel L (1986) Computation of the chi-square and Poisson distribution *SIAM J. Sci. Statist. Comput.* **7** 1022–1036

5 Arguments

1: RLAMDA – REAL (KIND=nag_wp) *Input*

On entry: the parameter λ of the Poisson distribution.

Constraint: $0.0 < \text{RLAMDA} \leq 10^6$.

2: K – INTEGER *Input*

On entry: the integer k which defines the required probabilities.

Constraint: $\text{K} \geq 0$.

3: PLEK – REAL (KIND=nag_wp) *Output*

On exit: the lower tail probability, $\text{Prob}\{X \leq k\}$.

4: PGTK – REAL (KIND=nag_wp) *Output*

On exit: the upper tail probability, $\text{Prob}\{X > k\}$.

5: PEQK – REAL (KIND=nag_wp) *Output*

On exit: the point probability, Prob$\{X = k\}$.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, RLAMDA ≤ 0.0.

IFAIL $= 2$

On entry, K < 0.

IFAIL $= 3$

On entry, RLAMDA $> 10^6$.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Results are correct to a relative accuracy of at least 10^{-6} on machines with a precision of 9 or more decimal digits, and to a relative accuracy of at least 10^{-3} on machines of lower precision (provided that the results do not underflow to zero).

8 Parallelism and Performance

G01BKF is not threaded in any implementation.

9 Further Comments

The time taken by G01BKF depends on λ and k. For given λ, the time is greatest when $k \approx \lambda$, and is then approximately proportional to $\sqrt{\lambda}$.

10 Example

This example reads values of λ and k from a data file until end-of-file is reached, and prints the corresponding probabilities.

10.1 Program Text

```
    Program g01bkfe

!     G01BKF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g01bkf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: peqk, pgtk, plek, rlamda
      Integer                          :: ifail, k
!     .. Executable Statements ..
      Write (nout,*) 'G01BKF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) '    RLAMDA     K     PLEK      PGTK      PEQK'
      Write (nout,*)

!     Loop over all data
d_lp: Do
        Read (nin,*,Iostat=ifail) rlamda, k
        If (ifail/=0) Then
!         All data processed
          Exit d_lp
        End If

!       Calculate probability
        ifail = 0
        Call g01bkf(rlamda,k,plek,pgtk,peqk,ifail)

!       Display results
        Write (nout,99999) rlamda, k, plek, pgtk, peqk
      End Do d_lp

99999 Format (1X,F10.3,I6,3F10.5)
    End Program g01bkfe
```

10.2 Program Data

```
G01BKF Example Program Data
  0.75     3     : RLAMDA, K
  9.20    12
 34.00    25
175.00   175
```

10.3 Program Results

```
 G01BKF Example Program Results

     RLAMDA     K     PLEK      PGTK      PEQK

      0.750     3   0.99271   0.00729   0.03321
      9.200    12   0.86074   0.13926   0.07755
     34.000    25   0.06736   0.93264   0.02140
    175.000   175   0.52009   0.47991   0.03014
```